```
!   RE-STORE "Corel_qp.prg"
!
!*******************************************************************************
!  "Quattro Pro example" for Tech Soft's MS Office function library
!
!  (c) Tech Soft GmbH 1999-2006                http://www.TechSoft.de
!  Author:           Sven Henze
!  Date:             03-Sep-2001
!  Last change:      02-May-2006
!
!*******************************************************************************
!
!
INTEGER Verbose=1           ! switch on/off verbose mode (0=off)
!
! Load the high level function library if not already loaded
IF NOT INMEM("Dde_lib_start") THEN LOADSUB ALL FROM "htbdde.lib"
!
!
! Now we call the custom SUB where the Office functions will be executed
! If you need more calling parameters either include them in the SUB definition
! or use COM blocks
!
CALL Dde_mainprog(Verbose)
!
!
! cleaning up SUB programs (in order to make sure that the most recent versions are loaded every time)
DELSUB Dde_lib_start TO END
!
END
!
SUB Dde_mainprog(OPTIONAL INTEGER Verbose)
!
! ************************************************************************
! *                                                                      *
! *  MS Office Interface for HTBasic: Corel Quattro Pro example          *
! *                                                                      *
! *  Version :  2.01.55                                                  *
! *  Author  :  Sven Henze, Tech Soft GmbH                               *
! *  Created :  03-Sep-2001                                              *
! *  Updated :  02-May-2006                                              *
! *                                                                      *
! *  Copyright (c) 1999-2006  Tech Soft GmbH. All Rights Reserved.       *
! *                           Email: HTB@TechSoft.de                     *
! *                           http://www.techsoft.de/dde.htm             *
! *                                                                      *
! ************************************************************************
!
! Description:
! ============
!
!   This example shows how to work with Corel Quattro Pro. It opens a Quattro Pro
!   template file, writes data into that file, reads out data from two cells
!   and creates a 3D diagram. Finally it saves the file under a new name.
!   The example has been created using Corel Quattro Pro 12 but should work
!   with earlier and newer versions of Quattro Pro too.
!
!   The entry point for your own DDE commands can be found under the label
!   named "Myown_dde_cmds". You can include any DDE code here using the
!   pre-allocated system variables (DDE channel numbers, application name etc.).
!   One example is given, it returns all available "Sysitems" of the application.
!
! Please note:
!   This program was written and optimized for HTBasic 8.3 and higher.
!   If you use an older version of HTBasic it might be neccessary to remove
!   or change some commands. The example does *not* work with HTBasic 7.x due
!   to some problems in the CSUB interface within HTBasic 7.x.
!
! define some variables
!
    DIM Doc_template$(1:10)[255],Doc_file$[255],App_exe$[255],Topic$[255],Tmp$[8192],Htbvers$[80],Os$[255],Question$[1024]
    INTEGER Printer,Btn,Sys_chn(1:5),Doc_chn(1:10),Result(1:10),App_quit,Language,Printlen
    DIM Ddecmd$[255],Info$[2048],Bookmark$[128],Doc password$[128],Sel$[255],Struct file$[255]
    DIM Top$[4096],Sname$[255],Value$[1024],App_name$[255],Htb_real$[255],Ret$(1:2)[4096]
    DIM G$[5],X$[5],Hl$[5],Hx$[5],App$[255],Crlf$[8],Wpwinvers$[20],I$[255],Colchar$[255],Newfile$[5]
    INTEGER Rows,Columns,Row_ofs,Col_ofs,I,J,Rv1,Cv1,Rv2,Cv2,Dummy,Xres
    !
    ON ERROR GOSUB Dde_err_handler        ! Global error handler
    !
    G$=CHR$(19)                                 ! character code for inverse printing
    X$=CHR$(16)                                 ! character code for normal printing
    Hl$=CHR$(30)                                ! character code for "highlighted" printing (different color)
    Hx$=CHR$(24)                                ! character code for "normal" printing (color)
    Crlf$=CHR$(10)&CHR$(13)
    !
    !
    INTEGER Verb_
    Verb_=0
    IF NPAR>=1 THEN
        IF Verbose>0 THEN Verb_=1
    END IF
    !
    Printlen=FNDde_defprintlen  ! define printlen for this session
    !
    ! define colors and other settings
    INTEGER Custom_colours
    Custom_colours=1                      ! use custom colours (set to 0 if you do not wish to change colour palette)
    !
    IF NOT FNDde_init_system(Custom_colours) THEN STOP
    !
    !
    ! define the application
    App$="QPW"     ! Corel Quattro Pro
```

```
        !
        ! Application quit behaviour
        ! Values:   0  = Do not close the application ever
        !           1  = Close the application after all DDE commands have been sent without any warning
        !           2  = Prompt the user to close the application
        !          >2  = App_quit=Timeout value (prompt the user to close the application until the given timeout has reached)
        !                If timeout occurs the dialog will be closed and the application will not be closed automatically
        App_quit=2
        !
        Topic$="QPW"
        App_name$="Quattro Pro"
        App_exe$=Topic$&".exe"
        !
        ! If the specified filename starts with a "*" a new file will be created and saved
        !Newfile$="*"              ! uncomment this if you want to create a new file
        !
        IF Newfile$="*" THEN
            ! If the specified filename starts with a "*" a new file will be created and saved
            Doc_template$(1)=Newfile$&SYSTEM$("MSI")&"\newfile.qpw"  ! Create new empty document
        ELSE
            Doc_template$(1)=SYSTEM$("MSI")&"\template.qpw"          ! Quattro Pro template
        END IF
        Doc_template$(2)=SYSTEM$("MSI")&"\template-output.qpw"       ! Quattro Pro output file
        !
        !
        IF Verb_>0 THEN CALL Dde_programhead("HTBasic - Corel Quattro Pro example")
        !
        CALL Dde_start(App_exe$,Topic$,Sys_chn(1),Language,Verb_)      ! start the DDE application defined in Topic$
        !
        IF NOT Sys_chn(1) THEN
            BEEP
            GOTO Finish
        END IF
        !
        ALPHA PEN 3
        IF Verb_>0 THEN OUTPUT CRT;" "&CHR$(20)&"Sending DDE commands to "&App_name$&X$
        ALPHA PEN 1
        !
        !
        ! first open a document in QPW
        Doc_chn(1)=FNDde_openfile(Topic$,Sys_chn(1),Doc_template$(1),Result(1),Doc_password$,Verb_)
        !
        IF NOT Doc_chn(1) THEN
            BEEP
            GOTO Finish
        END IF
        !
        !
Myown_dde_cmds: !
        !
        !=============================================================================================================
        !
        !  ******  Custom DDE programming starts here  ******
        !
        ! Todo: Write your own MS Office programming code here
        !
      !  CALL Dde_driver_info        ! shows info about DDE driver version and date
        !
      !  CALL Ddeexecute(Doc_chn(1),"{WindowQPW.Minimize}",Result(1))  ! Minimize QPW Window (makes Poke faster)
        !
Sheetdata: ! sheet data
        !
        Rows=6                       ! number of rows to write
        Columns=13                   ! number of columns to write
        Row_ofs=2
        Col_ofs=0
        !
        ! Data for Excel table
        DATA ,Jan,Feb,Mar,Apr,May,Jun,Jul,Aug,Sep,Oct,Nov,Dec
        DATA Sven,12.1234,11,10.043,9,8,7.106,6,5.0432,4,3.104,2,1
        DATA Dan,13,12.004,11,10.05,9,8,7,6.04,5,4.01,3,2.1
        DATA Pat,14,13.046,12,11,10.1,9,8.043,7,6.015,5,4.1,3.0206
        DATA Joerg,15,14.0418,13,12.1052,11,10.0832,9,8.1,7,6.1,5,4.03
        DATA Martin,16.031,15,14.06128,13,12.021,11,10.01,9,8,7,6,5.1
        !
        RESTORE Sheetdata
        !
        Dde_fprint("  -> Poking some values into the new sheet ",Printlen-2,Verb_)
        !
        Value$=Value$&"HTBasic DDE Test (current date and time: "&DATE$(TIMEDATE)&", "&TIME$(TIMEDATE)&")"
        ! Poke the data into the sheet
        CALL Ddepoke(Doc_chn(1),"A:A1",Value$,Result(1))
        !
        FOR I=1 TO Rows
            FOR J=1 TO Columns
                !
                I$="A:"&CHR$(64+J)&VAL$(I+Row_ofs)     !
                READ Value$
                !
                ! convert the HTBasic value into Windows locale format (decimal point!)
                Value$=FNDde_htbnum_2_xl$(Value$,Language,Xres)
                !
                ! Poke the data into the sheet
                CALL Ddepoke(Doc_chn(1),I$,Value$,Result(1))
            NEXT J
        NEXT I
        CALL Dde_show_status(Result(1),Verb_)
        !
        !
        ! now insert SUM formulas into the last row of each data column
        !
        FOR I=1 TO Columns
            Colchar$=CHR$(NUM("A")+I-1)
```

```
            IF I=1 THEN
                CALL Ddepoke(Doc_chn(1),"A:"&Colchar$&VAL$(3+Rows-1+Row_ofs),"Total",Result(1))
            ELSE
                Value$="{SELECTBLOCK "&Colchar$&"4.."&Colchar$&VAL$(3+Rows-1+Row_ofs)&"}"
                CALL Ddeexecute(Doc_chn(1),Value$,Result(1))
                CALL Ddeexecute(Doc_chn(1),"{SPEEDSUM}",Result(1))
            END IF
        NEXT I
        !
        ! now format the SUM row
        Colchar$=CHR$(NUM("A")+Columns-1)
        Value$="{SELECTBLOCK A"&VAL$(Rows+Row_ofs+2)&".."&Colchar$&VAL$(Rows+Row_ofs+2)&"}"
        CALL Ddeexecute(Doc_chn(1),Value$,Result(1))  ! select total line
        !
        ! format cells: Courier, 12 point, bold, color=4 (red)
        CALL Ddeexecute(Doc_chn(1),"{Setproperty Cell_Font;""Courier;12;1;0;0;4;0""}",Result(1))
        !
        ! Now insert a diagram. See the QPW macro help for more information.
        CALL Ddeexecute(Doc_chn(1),"{GraphNew Chart1}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range XAxisLabelSeries;(A:A4..A8)}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range LegendSeries;A:B3..M3}",Result(1))
        !
        ! create the diagram data ranges
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range 1;(A:B4..B8);1}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range 2;(A:C4..C8);1}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range 3;(A:D4..D8);1}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range 4;(A:E4..E8);1}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range 5;(A:F4..F8);1}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range 6;(A:G4..G8);1}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range 7;(A:H4..H8);1}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range 8;(A:I4..I8);1}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range 9;(A:J4..J8);1}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range 10;(A:K4..K8);1}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range 11;(A:L4..L8);1}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{Series.Data_Range 12;(A:M4..M8);1}",Result(1))
        !
        CALL Ddeexecute(Doc_chn(1),"{Series.Go}",Result(1))
        CALL Ddeexecute(Doc_chn(1),"{FloatCreate Graph;A:A11;105;135;A:G33;945;105;Chart1}",Result(1))
        !
        ! Insert diagram title
        CALL Ddeexecute(Doc_chn(1),"{GraphSettings.Titles ""Created by HTBasic for Windows"";"""";"""";"""";""""}",Result(1))
        !
        ! format the graph to 3D bargraph
        CALL Ddeexecute(Doc_chn(1),"{GraphSettings.Type ""3D Bar""}",Result(1))
        !
!        CALL Ddeexecute(Doc_chn(1),"{WindowQPW.Maximize}",Result(1))  ! Maximize QPW Window
        !
        !
        RANDOMIZE
        !
        ! Now retrieve some cells from the sheet
        !
        ! define the cells to be retrieved
        Rv1=3+INT((Rows-1)*RND)               ! random value R1 between 4 and 8
        Cv1=1+INT(Columns*RND)             ! random value C1 between 2 and 7
        Rv2=3+INT((Rows-1)*RND)               ! random value R2 between 4 and 8
        Cv2=1+INT(Columns*RND)             ! random value C2 between 2 and 7
        !
        ! Now read some values from Quattro Pro
        !
        ! Please note: If you request numbers you normally get numbers with a decimal point
        !              as specified in the system's locale configuration. This decimal point
        !              is "." for the United States and other English speaking countries
        !              but e.g. for German speaking it is ",". So we must convert the decimal
        !              point as well as the thousands separator back into a HTBasic readable
        !              format. Use the function FNDde_xl_2_htbnum$ for this purpose.
        !
        CALL Dderequest(Doc_chn(1),"A:"&CHR$(64+Cv1)&VAL$(Rv1),Ret$(1),Result(1))    ! get 1st value
        Htb_real$=FNDde_xl_2_htbnum$(Ret$(1),Language,Result(3))        ! do the conversion (if possible) !
        Dde_fprint("  -> Reading value of (A:"&CHR$(64+Cv1)&VAL$(Rv1)&"): "&Hl$&Htb_real$&Hx$&" ",Printlen,Verb_)
        CALL Dde_show_status(Result(1),Verb_)
        !
        CALL Dderequest(Doc_chn(1),"A:"&CHR$(64+Cv2)&VAL$(Rv2),Ret$(2),Result(1))    ! get 2nd value
        Htb_real$=FNDde_xl_2_htbnum$(Ret$(2),Language,Result(3))        ! do the conversion (if possible) !
        Dde_fprint("  -> Reading value of (A:"&CHR$(64+Cv2)&VAL$(Rv2)&"): "&Hl$&Htb_real$&Hx$&" ",Printlen,Verb_)
        CALL Dde_show_status(Result(1),Verb_)
        !
        !
        !  ******  Custom DDE programming ends here  ******
        !
        !=============================================================================================================
        !
        CALL Dde_focus2basic                        ! bring HTBasic window back to front
        !
        !
        IF Newfile$="*" THEN
            ! if the file was newly created simply save it under the given name
            Dummy=FNDde_savefile(Topic$,Doc_chn(1),Doc_template$(1),Result(1),2,"",Verb_)
        ELSE
            ! Save file under the name specified in Doc_template$(2). Ask for overwrites
            Dummy=FNDde_savefile(Topic$,Doc_chn(1),Doc_template$(2),Result(1),0,"",Verb_)
        END IF
        !
        Dde_enter_kbd(" Press <RETURN> to close the document(s).")
        !
        ! Close file and ask if file should be saved
        ! Doc_chn(1)=FNDde_closefile(Topic$,Sys_chn(1),Doc_template$(1),Result(1))
        !
        ! Close file, do not ask for save
        Dummy=FNDde_closefile(Topic$,Sys_chn(1),Doc_template$(1+Dummy),Result(1),1,Verb_)
        !
Finish:  !
        ! terminate all DDE connections and resources
```

```
        ! if variable "App_quit" is set to 1 or greater the application will be closed
        CALL Dde_quit(Sys_chn(1),Doc_chn(1),Result(1),App_quit,Topic$,Verb_,0,1)
        !
        ALPHA PEN 4
        DISP " Program finished."
        ALPHA PEN 1
        !
        SUBEXIT
        !
        ! Global error handler
Dde_err_handler:  !
        BEEP
        Dde_enter_kbd("**** Global Error Handler ****"&Crlf$&Crlf$&ERRM$)
        ERROR RETURN
        !
    SUBEND
```